
解   説  ・  ２

XPFortran 入門

永　井　　　亨

Ⅰ．予備知識

名古屋大学情報連携基盤センター全国共同利用システムのスーパーコンピュータは Fujitsu 

PRIMEPOWERHPC2500です。アプリケーションサーバも同一機種であるため，これらを

一体的に運用しています。XPFortranは，HPC2500上で複数の CPUを使って並列処理を行

うために必要な命令及びデータの分割や，CPU 間のデータ通信の命令を記述するための言語

で，Fortranに指示行を挿入する形式で表現します。並列処理プログラミングを説明する前に，

HPC2500及び XPFortran に関するいくつかの予備的な説明をしておくことにします。

１　ハードウェア構成

HPC2500のハードウェア構成を図１に示します。システムは 24ノードで構成される分散並

列型スカラ計算機で，ノード間は４GB/秒の高速クロスバネットワークで結合されます。各ノー

ドは共有メモリ型スカラ計算機で，24ノードのうちの 22ノードには 64個の CPU，残りの２ 

ノードには 128個の CPUを搭載し，全体では 1664CPUになります。各 CPU は理論最大性能

８Gflops1のスカラプロセッサです。ただし，これは乗算と加算が同時に実行された場合のピー

ク性能です。システム全体の理論最大性能は 12.5Tflopsになります。各ノードは主記憶 512GB

を搭載し，トータルで 12TBになります。磁気ディスク容量は 100TBです。24ノードのうちの

１ つは学内 LAN（NICE）に接続され，バッチ処理に加えて会話型処理のサービスも行ないます。

IOノード以外のノードは演算処理専用で，入出力は IOノードを経由して行ないます。

２　並列処理の実行イメージ

XPFortranで記述したプログラムの実行イメージを図２にしたがって説明します。３つのプロ

セスを使って並列処理を行なうとして，図２の左半分が XPFortran による並列処理プログラム

を，右半分が処理の流れをあらわしています。

図２のプログラム中の !xoclで始まる行が並列処理のための指示行です。実行の流れは以下

のようになります。

１．プログラム（ロードモジュール） が３つのプロセスのいずれか（図２ではプロセス１）にロー

１　１秒間に 8× 109回の浮動小数点演算ができることになります。

ドされて実行が始まります。したがって，開始直後はプロセス 1 だけが処理を行ないます（図

２では A の部分。この部分には通常の Fortranが記述されていると考えてください）。この

部分を逐次実行と呼びます。逐次実行を行なうプロセス（ここではプロセス１）をここでは

親プロセスと呼び，それ以外のプロセス（プロセス２とプロセス３）を子プロセスと呼ぶこ

とにします。

２．parallel region 文にさしかかったところでプログラムが他のプロセスに複写されて 2並列

処理が始まります。並列処理の開始直後はどのプロセスも同一の処理（図２では Bの部分）

を行なうことになります。この同一の処理を行なう部分を冗長実行と呼びます 3。

３．spread do文は doループを分割する命令です。この部分（spread do文から end spread 

do文まで）では，それぞれのプロセスが自身に割り当てられた doループの繰り返し範囲を

処理します。図２では１から300の繰り返し範囲を３等分して処理することを示しています。

４．spread region文から end spread region文までのところはそれぞれのプロセスが行なう

処理を個々に記述した部分です。図２ではプロセス１が処理 Eを，プロセス２が処理 Fを，

プロセス３が処理 Gをそれぞれ行ないます。

５．end parallel region文にさしかかると，並列処理は終了します。図２ではプロセス２及び

プロセス３ではプログラムの実行が終了し，プロセス１だけが実行を継続します（Ⅰの部分）。

図１　HPC2500 のシステム構成

２　これは fork 機能に似ています。
３　ちょっと考えると冗長実行は無駄な並列処理をしている部分のようにも思われますが，実際にプ
ログラムを書いてみると冗長実行を行なわなければならない部分はどうしても出てくるようです。

６．end文（または stop文）の実行によりプログラムが終了します。

parallel region文と end parallel region文の間で実行されるプログラム部分や spread do文（ま

たは spread region文）と end spread do文（または end spread region文）の間で実行される

プログラム部分をリージョンと呼びます。特に，parallel region文と end parallel region文の間

の部分をパラレルリージョンと呼びます。リージョンは入れ子になっていてもよくて，パラレル

リージョンはすべてのリージョンの親リージョンとなります。これを図２で説明しますと，B か

ら H までのそれぞれの部分がリージョンを形成していて，図の右半分はそれぞれのリージョン

がどのプロセス上で実行されるかをあらわしていることになります 4。また，すべてのプロセス

が同一リージョンを実行することが冗長実行であることになります。

図２中の処理の流れの方向をあらわす矢印を時間軸としてみれば，各プロセスはそれぞれの

リージョンの実行を同時に開始して，同時に終了していることになります。これは，プロセス間

で自動的にバリア同期を取るためです。バリア同期とは，並列処理しているプロセスが同期を取

るべき場所（バリアポイント）で実行を停止し，すべてのプロセスがバリアポイントに達したと

ころで実行を再開する同期方式です。バリア同期を取る必要がなければ，それを指示することも

できます。

図２　並列処理の実行イメージ

４　正確に言うと，B，D及び Hは同一リージョン（パラレルリージョン）にあります。また，
spread do 文から end spread do 文までの部分に関しては，doループの繰り返しの１つ１つがそ
れぞれリージョンを形成するので 300個のリージョンを三等分してそれぞれのプロセス上で実行
することになります。

３　グローバル変数とローカル変数

通常，１つのプロセスは固有のメモリ空間をもち，他のプロセスからアクセスできませんが，

XPFortranではすべてのプロセスが共有するメモリ空間をもつことができます。この空間をグ

ローバル空間と呼び，グローバル空間上に配置された変数（または配列）をグローバル変数（ま

たはグローバル配列）と呼びます。図３に示すように，グローバル変数は DTU（data transfer 

unit）及びクロスバネットワークを介してすべてのプロセスからアクセスできます。また，グロー

バル変数はパラレルリージョンの内外でアクセスできます。

一方，グローバルとして宣言されなかった変数（または配列）をローカル変数（またはローカ

ル配列）と呼び，図 3 に示すように各プロセス固有の空間（ローカル空間）上に割り付けられます。

ローカル変数は他のプロセスからはアクセスできません。ローカル変数には２種類あって，複数

のプロセスに分割して割り当てられるものを分割ローカル配列とよび，それぞれのプロセス上に

分割されずに宣言どおり割り当てられるものを重複ローカル変数（または重複ローカル配列）と

呼びます。重複ローカル変数はパラレルリージョンの外では親プロセス上の値だけが使え，パラ

レルリージョンにさしかかったところで親プロセス上の値が子プロセス上に複写されます。分割

ローカル配列はパラレルリージョン内だけでアクセスできます。

プログラム表記上はグローバル変数の引用や定義は通常の変数（ローカル変数）の場合と同様

ですが，実際にはプロセス間の通信が発生しますのでアクセス速度はローカル変数の場合に比べ

て遅くなります。

４　プロセス間のデータ転送

グローバル変数をアクセスする場合にクロスバネットワークを介したプロセス間の通信が発生

図３　グローバル変数とローカル変数

します。クロスバネットワークの最大データ転送速度は４GB/秒です。この速度は現在ある並

列計算機のプロセス間の通信速度としては速い方なのですが，それでも主記憶とCPUの間のデー

タ転送速度と比べると決して速くはありません。また，４GB/秒の速度が常に出るわけではな

くて，転送速度は転送するデータ量に依存します。実際の転送時間には転送のための初期化など

に要するオーバーヘッドが含まれますのでデータ量が少ない転送を頻繁に行なうことは避け，で

きるだけまとめて転送するように心掛けましょう。

Ⅱ．並列処理プログラムの記述法

先ほども少し説明しましたが，XPFortranではソースプログラムに XPFortran用拡張最適化

制御行と呼ばれる並列処理のための指示行を挿入します。これは第１桁～第５桁が !xoclまた

は *xoclであるもので（以下，xocl行と呼びます），第６桁は空白，そして第７桁以降に指示

する内容（XPFortran構文と呼びます）を記述します 5。指示する内容が複数行にまたがる場合

には，行の最後に &を付けてつぎの行に継続します。xocl行は，記述される文の種類によって

XPFortran宣言構文と XPFortran実行構文とに分類されます。xocl行で指定できる文を表１に

示します。xocl行に関して以下のような制約があります。

•表１に示した２種類の構文を Fortranプログラム上で指定できる位置は，（Fortranの）宣

言文及び実行文が指定できる位置と同じです。

• xocl行で定義される名前（後述するプロセッサグループ名など）は xocl行以外では使用で

きません。

１　並列処理プログラム例１

まず，簡単な並列処理プログラムを作ってみることにします。ここではパラメータや入力デー

タだけを変えて，全く同じ計算を各プロセスで行なわせるという並列処理を考えます。これは最

５　xocl行の先頭は !または＊で始まりますから，他の Fortran処理系では注釈行とみなされます。
また，xocl行では小文字と大文字の区別はありませんので，本稿ではすべて小文字で記述してあ
ります。

XPFortran 宣言構文 XPFortran 実行構文

processor文
proc alias文
index partition文
local文
global文
subprocessor文
resident文
commonid文

parallel region文
end parallel region文
spread region文
end spread region文
spread do文
end spread do文
spread move文
end spread move文
spread assignment文
end spread assignment文

unify文
overlapfix文
move wait文
broadcast文
barrier文
lockon文
end lockon文
pass by local文

表１　xocl 行で指定できる文

も単純な（最も粒度の粗い）並列化といえるでしょう。例題としてつぎのような２次元の拡散方

程式を適当な初期条件及び境界条件の下で差分法で解くことにします。

( )  （１）

式（１）を，時間を前進差分，空間を中心差分でそれぞれ離散化します。x=i x，y=j y として，

関数 u（x，y，t）の時刻 t=k tにおける格子点（i，j）上の値を であらわせば，差分方程式

はつぎのようになります。

 （２）

ただし，

 （３）

です。

計算領域を 0 x  xmax及び 0 y ymaxとして， と tの値の組をいくつか与え，それぞれの

組について差分方程式（２）を解くことにします。xmax，ymax， x， yの値は固定します。並列

処理プログラムを図４に示します。ただし，図４では初期条件及び境界条件をそれぞれ設定する

サブルーチン init 及び boundは省略してあります。以下で，図４の並列処理プログラムにつ

いて説明します。

１行目 いくつかの整定数名を定義します。

２行目 複数のプロセスをまとめて扱う場合，そのまとまりをプロセッサグループと呼び

ます。processer文はプログラムが使用するすべてのプロセスから成るプロセッ

サグループの名前と使用するプロセスの総数 6を指定します。ここでは，mproc

というプロセッサグループ名で５台のプロセスを使用することを宣言していま

す。

３行目 index partition文は命令やデータをどのように分割するかを定義し，それを名前

として宣言します。ここでは名前（分割指定名と呼びます）を indと付けてい

ます。分割を指定する項目として，プロセッサグループ名，インデックス範囲

（index=の部分），分割方法（part=の部分）等があります。インデックス範囲

には，doループを分割する場合には do変数の初期値及び終値を，配列を分割

する場合には寸法の下限及び上限を指定します 7。分割方法には，均等（band），

循環（cyclic），不均等の３つがあります。分割方法を省略した場合の標準値

は bandです。この index partition 文は 15～ 27行目の doループの分割を指定

６　正確にはプロセスの総数ではなくて形状と呼びます。配列の宣言と同様に多次元にすることも
できるためです。図４の場合は１次元プロセッサグループということになります。
７　インデックス範囲の下限が１のときには 1:の部分を省略できます。

01:       parameter (npe=5,ncase=10,imax= 100,jmax= 100,kstep= 50)
02: !xocl processor mproc(npe)
03: !xocl index partition ind=(mproc,index=1:ncase,part=cyclic)
04: c
05:       dimension u(0:imax,0:jmax),wk(0:imax,0:jmax)
06:       dimension alpha(ncase),dt(ncase)
07:       data xmax,ymax/1.0d0,1.0d0/
08:       dx=xmax/dble(imax)
09:       dy=ymax/dble(jmax)
10:       t=0.0d0
11:       read(1,500) (alpha(i),dt(i),i=1,ncase)
12: 500   format(2d10.2)
13: c
14: !xocl parallel region
15: !xocl spread do /ind
16:       do icase=1,ncase
17:         sx=alpha(icase)*dt(icase)/dx**2
18:         sy=alpha(icase)*dt(icase)/dy**2
19:         call init(imax,jmax,u,xmax,ymax,alpha(icase),t)
20:         do k=1,kstep
21:           call bound(imax,jmax,u,xmax,ymax,alpha(icase),t)
22:           call diffus(imax,jmax,u,sx,sy,wk)
23:           t=t+dt(icase)
24:         end do
25:         write(icase+10) alpha(icase),dt(icase),u
26:       end do
27: !xocl end spread do
28: !xocl end parallel region
29:       end
30: c
31:       subroutine diffus(imax,jmax,u,sx,sy,wk)
32:       dimension u(0:imax,0:jmax),wk(0:imax,0:jmax)
33:       do j=1,jmax-1
34:         do i=1,imax-1
35:           wk(i,j)=sx*u(i+1,j)+(1.0d0-2.0d0*sx-2.0d0*sy)*u(i,j)
36:      *     +sx*u(i-1,j)+sy*u(i,j+1)+sy*u(i,j-1)
37:         end do
38:       end do
39: c
40:       do j=1,jmax-1
41:        do i=1,imax-1
42:          u(i,j)=wk(i,j)
43:        end do
44:       end do
45:       end

…

図４　並列処理プログラム例１

するためのものですが，cyclicを指定すると，doループの１回目をプロセス

１で，２回目をプロセス２で，. . . ，５回目をプロセス５で，６回目をプロセス

１で，. . . ，というようにサイクリックに処理が割り当てられます。

5～ 12行目 配列の宣言とパラメータの設定を行ないます。配列 wkはサブルーチン diffus

で使う作業領域です。 と tは装置参照番号１から読み込みます（11行目）。こ

の部分は逐次実行ですから親プロセスだけが処理を行ないます。

14行目 parallel region文にさしかかると並列処理が始まります。親プロセス上のプログ

ラムが子プロセス上に複写され，並列処理の開始直後は各プロセスの変数や配列

はそれぞれ同じ値を持ちます。だたし，この段階では配列 uなどは未定義のま

まです。図４中にあらわれるすべての変数及び配列は各プロセス上に重複して割

り当てられるので重複ローカル変数となります。

15行目 spread do文は doループを分割する命令です。３行目で宣言した分割指定 ind

にしたがって，直後の doループ（16～ 26行目）を分割して処理します。

16～ 26行目 入力した と tの各組について，指定した計算ステップ数（kstepの値）まで

計算して結果をファイルに出力します。この部分は各プロセスが与えられた と
tの値を使って同一の処理を行ないます。

27行目 end spread do文８は 15行目の spread do文に対応していて，分割した doルー

プの直後に置きます。

28行目 end parallel region文９は並列処理の終了を指示します。

31～ 45行目 サブルーチン diffusは差分式（２）を計算する部分です。diffusには xocl

行は挿入されていません 10。

図４に示したように，パラメータや入力データだけを変えて各プロセスで同一の処理を行なう

ならば，xocl 行を挿入する箇所も少なく，比較的容易に並列化できます。また，プロセス間の

通信もほとんど発生しませんから処理速度は使用するプロセス数にほぼ比例して伸びていくこと

が期待できます。

２　配列の分割

つぎに，配列を分割して並列処理する場合のプログラミングを簡単な例を使って説明します。

まず，寸法 30の１次元配列 a，b，cをローカル配列として宣言し，それぞれの配列を３等分し

て３台のプロセスに分割して割り当てることを考えます。プログラムはつぎのようになります。

８　end spread do の doは省略可能です。
９　end parallel regionの regionは省略可能です。
10　図４では省略されているサブルーチン init及び boundにも xocl行は挿入されていません。

!xocl processor p(3)

!xocl index partition ind=(p,index=1:30,part=band)

      dimension a(30),b(30),c(30)

!xocl local a(/ind),b(/ind),c(/ind)

　 　 　 　 　 　 　 　 　 　 …

!xocl spread do /ind

      do i=1,30

        c(i)=a(i)+b(i)

      end do

!xocl end spread do

local文はローカル変数であることを宣言します 11。ローカル配列を分割する場合には（分割ロー

カル配列）分割指定 indを使って，a(/ind)のような形式で指定します。分割方法は bandで

すから，インデックス範囲を均等に分割します。したがって，図５に示すようにプロセス１上

には a(1)～ a(10)が，プロセス２上には a(11)～ a(20)が，プロセス３上には a(21)～
a(30)がそれぞれ配置されます。配列b及びcも同様に分割して配置されます。後に続く doルー

プの分割でも spread do文で分割指定 indを使っています。この分割によって，プロセス１上

では do変数 iの値が１～ 10 の範囲を，プロセス２上では 11～ 20の範囲を，プロセス３上で

は 21～ 30の範囲をそれぞれ処理することになります。

この例では，a，b，cはローカル配列ですから，例えば，プロセス 1 から a(11)～ a(30)のデー

タをアクセスすることはできません。もし，他のプロセス上にあるデータをアクセスする必要が

あるならば，上のプログラムの宣言部分をつぎのように書き換えます。

図５　ローカル配列の分割

11　local文にも global 文（後述）にもあらわれない変数や配列はすべて重複ローカル変数になり
ます。

!xocl processor p(3)

!xocl index partition ind=(p,index=1:30,part=band)

      dimension a(30),b(30),c(30)

      dimension ag(30),bg(30),cg(30)

!xocl local a(/ind),b(/ind),c(/ind)

!xocl global ag,bg,cg

      equivalence(a,ag),(b,bg),(c,cg)

global文はグローバル変数であることを宣言します 12。よって，agはグローバル配列になります。

また，equivalence文によってグローバル配列 agはローカル配列 aと同一の記憶領域を共有す

ることになります。したがって，global文には分割指定がありませんが 13，図６に示すように配

列 agは３等分されて各プロセス上に配置されます。配列 bg，cgも同様です。このようにすると，

例えば，プロセス１上の計算において

　　　　c(1)=ag(11)+bg(30)

といった記述が可能になります 14。しかし，ローカル変数と同一の記憶領域を共有していても
ag，bg，cgはグローバル変数ですから，アクセス速度はローカル変数に比べて遅くなります。よっ

て，できる限りローカル変数へのアクセスだけで計算を済ませるように配慮する必要があります。

つぎに，下のような doループを３等分して，３台のプロセスを使って並列処理することを考

えます。

図６　ローカル配列とグローバル並列の記憶領域の共有

12　global文に指定できる変数の型は基本整数型，８バイト整数型，基本論理型，８バイト論理型，
基本（単精度）実数型，倍精度実数型，4倍精度実数型，基本（単精度）複素数型，倍精度複素数型，
４倍精度複素数型です。

13　!xocl global ag(/ind)，bg(/ind)，cg(/ind) のように書くこともできます。
14　ag(11)及び bg(30)はそれぞれ a(11)及び b(30)と同じ記憶領域を占めています。しかし，
c(1)=a(11)+b(30)という記述は許されません。

dimension a(30),b(30)

　 　 　 　 　 …

do i=2,29

b(i)=a(i-1)+a(i)+a(i+1)

end do

配列 a及び bは先ほどと同様に３等分して３台のプロセス上に分割ローカル配列として配置し

ます。また，doループは，プロセス１上では do変数 iの値が２～ 10の範囲を，プロセス２上

では 11～ 20の範囲を，プロセス３上では 21～ 29の範囲をそれぞれ分割して処理することにな

ります。ところが，変数 iの値が 10のときの計算では a(i+1)すなわち a(11)はプロセス２

上にありますから，プロセス１からはアクセスできません。これは変数 iの値が 11，20，21 の

ときにも同様なことが起こります。このように配列を複数のプロセス上に分割することによって

「泣き別れ」になる部分のことを袖と呼びます。袖を含む計算では，先ほどのように equivalnce

文を使ってグローバル配列と結合してやればよいのですが，これに加えてもう一工夫します。

上の例に xocl行を挿入したプログラムは図７のようになります。以下で，図７について説明

します。

01: !xocl processor p(3)
02: !xocl index partition ind1=(p,index=1:30,part=band,overlap=(1,1))
03: !xocl index partition ind2=ind1(overlap=(0,0))
04:       dimension a(30),ag(30),b(30)
05: !xocl local a(/ind1),b(/ind2)
06: !xocl global ag
07:       equivalence (a,ag)

　 　 　 　 　 　 　 　 …

08: !xocl overlapfix(a) (id)
09: !xocl move wait (id)
10: !xocl spread do /ind2
11:       do i=2,29
12:         b(i)=a(i-1)+a(i)+a(i+1)
13:       end do
14: !xocl end spread do

図７　袖付きローカル配列の分割例

図８　ローカル配列 aとグローバル配列 agの分割のイメージ

２行目 分割指定 ind1の項目の中の overlap=の部分が袖をどれだけ伸ばすかを指定す

るものです 15。overlap=（l，r）は，インデックス範囲を指定された分割方法 16

で分割した結果に，インデックスが減少する方向へ l個，また，インデックスが

増加する方向へ r個伸ばした範囲を割り当てることをあらわしています。よって，

図８に示すように，ローカル配列 aはプロセス１上には a(1)～ a(11)が，プロ

セス２上には a(10)～ a(21)が，プロセス３上には a(20)～ a(30)がそれぞ

れ割り当てられます。袖部分は隣接するプロセス上に重複して割り当てられるこ

とになります。グローバル配列 agは袖付きの分割ローカル配列 aと equivalence

文で結合しています（７行目）が，図８のように agは袖を持ちません。図８中

の各プロセス上の配列 aの網掛けをした部分が袖になります。また，各プロセス

上の配列 aと agを点線で結んだ部分が互いに共有する記憶領域になります。

３行目 すでに定義した分割指定（ここでは ind1）の一部を変更して新たな分割指定（こ

こではind2）を宣言する方法です。分割指定ind2は袖なしの均等分割になります。

８行目 プロセス１上の配列要素 a(11)（これは袖です）とプロセス２上の a(11)（こ

れは配列要素 ag(11)と記憶領域を共有しています）は異なるプロセス上に配

置されていますから，計算の途中のある時点ではどちらか一方が未定義であった

り，互いに異なる値を持ったりする可能性があります 17。この不整合を解消する

ための命令が overlapfix文です。overlapfix文は，指定された袖付き分割ローカ

ル配列（ここでは配列 a）の袖部分の値を，equivalence文で結合されたグロー

バル配列（ここでは配列 ag）の同じ添字を持つ要素の値に一致させるためにデー

タ転送の開始を指示します 18。これを図８を使って説明しますと，配列 agの斜

線を引いた部分（要素）の値を矢印で結んだ先の配列 aの袖部分に複写するこ

とになります。これによって，例えばプロセス１上の a(11)の値はプロセス２

上の ag(11)の値に置き換わります。

９行目 overlapfix文は袖部分の値の置き換えの開始を指示する命令ですから，データ

転送の終了を待たずに後続の文の実行に移ります。したがって，overlapfix文の

直後に袖の値を参照する文を書いてもまだデータ転送が終了していない可能性

があります。データ転送が完了したことを確認する命令がmove wait文です 19。

overlapfix文とこれに対応するmove wait文を識別するために，２つの文に同じ

15　袖付きの分割は，local 文，global 文，spread do 文，spread move 文，spread assignment 文
で指定できます。

16　part=cyclicとした場合には袖は指定できません。
17　これは実際に起こります。
18　当然，データ転送を開始する時点では equivalence 文で結合されたグローバル配列の持つ値が
正しい値であることを前提にしています。

19　move wait文の位置でバリア同期がとられます。

プロセッサ間データ転送識別名（ここでは id）を付けます 20。

10～ 14行目 spread do文では分割指定 ind2を指定します（ind1を指定するのは誤りです）。

doループは分割されて，プロセス１上では do変数 iの値が２～ 10の範囲を，

プロセス２上では 11～ 20の範囲を，プロセス３上では 21～ 29の範囲をそれ

ぞれ処理することになります 21。

３　並列処理プログラム例２

配列を分割した並列処理プログラムの例題として，先ほどの２次元拡散方程式（１）を差分法

で解くことを考えます。ただし，入力となるパラメータは１組だけにして，計算領域を分割して

処理します。並列処理プログラムの主プログラム部分を図９に，サブルーチン diffusの部分

を図 10にそれぞれ示します。

以下で，図９及び図 10の並列処理プログラムについて説明します。

４～５行目 この index partition文は配列 uの分割を指定します。４行目の最後の &はこ

の xocl行がつぎの行に継続することをあらわしています。インデックス範囲

を index=0:jmaxとしていますから，配列 uの２次元目（ y軸方向）を分割

します。分割方法は bandですから均等分割になりますが，配列 uの２次元目

の寸法は 101なのでプロセス数の５で割ると端数がでます。この場合には，商

に１を足した要素数を順に割り当てていって，最後に残りの要素数をそのつぎ

のプロセスに割り当てます。101を５で割ったときの商は 20ですから，最初の

４プロセスではそれぞれ 21要素，最後の１プロセスでは 17要素となるように

分割されます 22。この分割指定では袖も指定していますが，差分式（２）から袖

の指定が必要になることはお分かり頂けると思います。

７行目 分割する配列が多次元の場合，この local文にあるように分割しない次元（この

場合１次元目）にはコロン（:）を置きます。

９行目 ローカル配列 uとグローバル配列 ugを equivalence 文で結合します。

10行目 グローバル配列と equivalence文で結合されたローカル配列は common文に指

定できないので，グローバル配列 ugを指定してプログラム単位間のデータの結

合を行ないます 23。

20　データ転送の開始と終了の確認が２つの文に分かれていますから，２つの文の間にデータ転
送とは無関係な処理を行なう命令を書き込むこともできます。ただし，通常は overlapfix 文と
movewait文は同じプログラム単位内になければなりません。

21　spread do文に分割指定 ind1 を指定すると，各プロセス上で袖を含む繰り返し範囲を処理する
ことになります。

22　4× (20+1)+17=101です。
23　一般には，common文にはローカル変数，グローバル変数のいずれも指定できます。ただし，
重複ローカル変数，分割ローカル配列，グローバル変数が１つの共通ブロックに混在することは
できません。

01:       parameter(kstep= 50)

02:       parameter (npe=5,imax= 100,jmax= 100)

03: !xocl processor mproc(npe)

04: !xocl index partition ind1=(mproc,index=0:jmax, &

05: !xocl        part=band,overlap=(1,1))

06:       dimension u(0:imax,0:jmax),ug(0:imax,0:jmax)

07: !xocl local u(:,/ind1)

08: !xocl global ug

09:       equivalence (u,ug)

10:       common /data/ ug

11: c

12:       data xmax,ymax/1.0d0,1.0d0/

13:       dx=xmax/dble(imax)

14:       dy=ymax/dble(jmax)

15:       dt=2.0d0

16:       alpha=1.0d-5

17:       sx=alpha*dt/dx**2

18:       sy=alpha*dt/dy**2

19:       t=0.0d0

20: c

21: !xocl parallel region

22:       call init(xmax,ymax,alpha,t)

23:       do k=1,kstep

24:         call bound(xmax,ymax,alpha,t)

25: c

26: !xocl overlapfix(u) (id)

27: !xocl move wait (id)

28: c

29:         call diffus(sx,sy)

30:         t=t+dt

31:        end do

32: c

33: !xocl end parallel region

34:       write(1) ug

35:       end

図９　並列処理プログラム例２－１（主プログラム）

36:       subroutine diffus(sx,sy)

37:       parameter (npe=5,imax= 100,jmax= 100)

38: !xocl processor mproc(npe)

39: !xocl subprocessor sproc=mproc(1:npe)

40: !xocl index partition ind1=(sproc,index=0:jmax, &

41: !xocl         part=band,overlap=(1,1))

42:       dimension u(0:imax,0:jmax),ug(0:imax,0:jmax)

43: !xocl local u(:,/ind1)

44: !xocl global ug

45:       equivalence (u,ug)

46:       common /data/ ug

47: c

48: !xocl index partition ind2=(sproc,index=0:jmax,part=band)

49:       dimension wk(0:imax,0:jmax)

50: !xocl local wk(:,/ind2)

51: c

52: !xocl spread do /ind2

53:       do j=1,jmax-1

54:         do i=1,imax-1

55:           wk(i,j)=sx*u(i+1,j)+(1.0d0-2.0d0*sx-2.0d0*sy)*u(i,j)

56:      *      +sx*u(i-1,j)+sy*u(i,j+1)+sy*u(i,j-1)

57:         end do

58:       end do

59: !xocl end spread do

60: c

61: !xocl spread do /ind2

62:       do j=1,jmax-1

63:        do i=1,imax-1

64:          u(i,j)=wk(i,j)

65:        end do

66:       end do

67: !xocl end spread do

68:       end

図 10　並列処理プログラム例２－２（サブルーチン diffus）



名古屋大学情報連携基盤センターニュース　Vol.５,　No.２－2006. 5－144

12～ 19行目 パラメータの設定。

21行目 パラレルリージョンの始まり。

26～ 27行目 計算ステップごとに，サブルーチン diffusを呼び出す前に分割ローカル配列 u

の袖部分の値をグローバル配列 ugの対応する要素の値に一致させます。

29行目 サブルーチン diffusを呼び出します。配列 u（正確には ug）は共通ブロック

でサブルーチン側に引き渡されます 24ので，図４の並列処理プログラム例１の

中の diffusと比べて引数の並びが異なっています。

33行目 パラレルリージョンの終り。

34行目 計算結果をファイルに出力します。write文にグローバル配列 ugを指定してい

ます。グローバル配列はパラレルリージョンの外側（逐次実行時）でもアクセス

できます。

36行目 サブルーチン diffusの始まり（ここから図 10を御覧ください）。図４の並列

処理プログラム例１の場合と異なり，サブルーチン内部も分割して処理するので

xocl行の挿入が必要になります。

37～ 46行目 この宣言部分は主プログラムの宣言部分（図９の２～ 10行目）とよく似てい

ますが，39行目に主プログラムにはない subprocessor文があります。実引数

と仮引数の関係に類似して，プロセッサグループには実プロセッサグループと

仮プロセッサグループがあります。subprocessor文は仮プロセッサグループ

の名前とその形状を宣言します。39行目では，仮プロセッサグループ sproc

は実プロセッサグループ mprocの１～５番目のプロセスと結合することを

あらわしています 25。このサブルーチン内では，40～ 41行目や 48行目の index 

partition文にあるようにプロセッサグループには仮プロセッサグループ名を使い

ます。

48～ 50行目 作業領域 wkを（袖なしの）分割ローカル配列として宣言しています。

52～ 59行目 spread do文の直後の doループ（ここでは do変数 jに関するループ）が分割

されます。分割指定は ind1ではなくて ind2を指定していることに注意してく

ださい。

図９及び図 10には示されていませんが，サブルーチン init及び boundにも diffusと同

24　共通ブロックを使わずに，分割ローカル配列やグローバル配列を引数に指定することも可能で
す。ただし，この場合には分割ローカル配列 uをサブルーチン側で equivalence文を使ってグロー
バル配列と結合させることはできなくなります。Fortranの規約上，仮引数に指定した変数名や
配列名を equivalence文に指定することができないからです。

25　38行目の processor文にあるように mprocはプログラムで使用するすべて（５つ）のプロセ
スで構成されていますから，結果的に mprocというプロセッサグループ名に sprocという別名
を付けたことにもなります。このような指定の仕方をプロセッサグループ固定手続と呼びます。
他にプロセッサグループ大きさ固定手続とプロセッサグループ大きさ引継ぎ手続があります。

様な xocl行が挿入されています 26。

この例題では２次元配列 uの２次元目を分割していますが，多次元配列の分割の仕方は全体

の処理速度に影響を与えることがあります。プロセス間の通信を少なくすることと，各プロセス

の処理性能を引き出すことを考えて命令及び配列を分割することが基本でしょう。

４　グローバル変数・ローカル変数間のデータ転送

すでに説明したようにグローバル変数はどのプロセスからもアクセスできますが，プロセス間

の通信が発生しますからアクセス速度はローカル変数に比べて遅くなります。アクセスするグ

ローバル変数が同じノードの主記憶上にある場合にはクロスバネットワークを介した通信は起

きませんが，通常の主記憶へのアクセスとは異なる 27ためやはりアクセス速度は遅くなります。

このためグローバル変数を使って計算などの処理を行なう場合には，グローバル変数を一旦ロー

カル変数に転送し，そのローカル変数を使って処理を行なう方が効率的です 28。グローバル変数

とローカル変数との間のデータ転送には spread move文を使います 29。

４. １　データ転送例 1

２次元のグローバル配列をローカル配列に複写するにはつぎのようにします。

!xocl processor p(2)

!xocl index partition ind=(p,index=1:100,part=band)

      dimension a(100,100),b(100,100)

!xocl global a(:,/ind)

!xocl local b(:,/ind)

　 　 　 　 　 …

!xocl spread move /ind,:

      do j=1,100

        do i=1,100

          b(i,j)=a(i,j)

        end do

      end do

!xocl end spread move

26　特に宣言部分（図 10では 37～ 46行目）は各サブルーチンで共通になりますから，include文
を使うと便利です。

27　グローバル変数へのアクセスでは DTUを経由します。
28　ここではグローバル変数とローカル変数とを equivalence文で結合することができない場合や
結合できても不都合が生じる場合を考えています。

29　spread move文を使うとデータは一括して転送されます。

グローバル配列 a及びローカル配列 bはいずれも２次元目が分割されていますから，図 11に示

すように２つのプロセス上に配置されます。これを図 11のようにグローバル配列a(1:100,1:50)

の値をプロセス１上の分割ローカル配列 b(1:100,1:50)に，a(1:100,51:100)の値をプロセ

ス２上の b(1:100,51:100)にそれぞれ転送するプログラムです。

spread move文から end spread move文 30までの間にある doループのそれぞれについてどの

ように分割するかを spread move文で指定します 31。この例では do変数 jのループを分割し，

do変数 iのループは分割しませんから，外側にある doループから順番に対応させて /ind,:

と指定します。分割指定と分割指定の間はコンマ（,）で区切ります。コロンを指定すると対応

する doループは分割されません。これによって例えば，プロセス１上では do変数 jが１～ 50

の範囲，iが１～ 100の範囲を処理することになります 32。

上のプログラムの最後の end spread move 文を

!xocl end spread move(id)

!xocl move wait(id)

図 11　spread move 文による転送（その１）

30　end spread moveの moveは省略できます。
31　spread do 文は直後の do ループだけを分割します。
32　この例ではグローバル配列 aとローカル配列 bの分割の方法は同じですからプロセス間の通信
は起こらず，同じプロセス上でのデータ転送になります。

と記述することもできます。これはoverlapfix文と同様に spread move文で転送の開始を指示し，

転送の完了は end spread move文で指定したプロセッサ間データ転送識別名（この場合 id）と

同じ名前が指定されたmove wait文によって確認することを指定します 33。上のプログラムのよ

うに end spread move文で（id）を省略すると，対応するmove wait文は end spread move文

の直後にあるものとみなされます。

ローカル配列 bからグローバル配列 aへデータ転送する場合には上の例の代入文の左辺と右

辺を入れ換えて，
a(i,j)=b(i,j)

とします。

配列代入を使って表現する場合には，上のプログラムの spread move文から end spread 

move文までを，

!xocl spread move :,/ind

      b=a

!xocl end spread move

とします。spread move文の分割指定の :,/indが上のプログラムと反対になっていることに

注意してください。これはローカル配列 bの分割指定に合わせているためです。

４. ２　データ転送例 2

この例も２次元のグローバル配列をローカル配列に複写する例です。ただし，グローバル配列
aは２次元目が，ローカル配列 bは１次元目がそれぞれ分割されています。

!xocl processor p(2)

!xocl index partition ind=(p,index=1:100,part=band)

      dimension a(100,100),b(100,100)

!xocl global a(:,/ind)

!xocl local b(/ind,:)

　 　 　 　 　 …

!xocl spread move :,/ind

      do j=1,100

        do i=1,100

          b(i,j)=a(i,j)

33　end spread move文とmove wait文の間にデータ転送とは無関係な処理を行なう命令を書き込
むこともできます。

        end do

      end do

!xocl end spread move

この例では図 12のような転送が行われます。例えば，プロセス１上では do変数 jが１～ 100 

の範囲，iが１～ 50の範囲を処理することになります 34。この例と前の例とを比べるとお分か

りいただけるかと思いますが，処理を割り当てられたプロセスが，処理する範囲にあるすべて

の配列要素にアクセスできる必要がありますから，ローカル配列の分割指定に合わせて spread 

move文の分割指定をすることになります。

４. ３　データ転送例 3

グローバル配列を重複ローカル配列に複写する場合にはつぎのようになります。

!xocl processor p(2)

!xocl index partition ind=(p,index=1:100,part=band)

      dimension a(100,100),b(100,100)

!xocl global a(:,/ind)

　 　 　 　 　 …

!xocl spread move

      do j=1,100

図 12　spread move 文による転送（その２）

34　この場合にはプロセス間の通信が発生します。

        do i=1,100

          b(i,j)=a(i,j)

        end do

      end do

!xocl end spread move

この例では図 13のような転送が行われます。aはグローバル配列，bは重複ローカル配列です。

この場合のように spread move文に分割指定が省略されている場合にはすべてのプロセスがす

べての doループの範囲を処理します 35。

一般に，spread move文と end spread move文の間にあらわれる doループ中の代入文は

a(i1,i2, . . .,im, . . .)=b( j1, j2, . . ., jn, . . .)

の形で，以下のような制約があります。

１．配列 a及び bは同じ型の配列でなければなりません。

２．添字 im，jn はつぎの３つの形式のいずれかでなければなりません。

• b（do 変数を含まない整数式）

•［+または－］［a*］j［+bまたは－ b］（do 変数を含む整数式）

• list(j)（リストベクトル）

ただし，jは do変数，a及び bは do変数を含まない整定数，整数型変数または整数式のい

1 50

1

100

1 50
1

100

a

b

51 100

51 100

1

100

1 50
1

100

51 100

����1 ����2

図 13　spread move 文による転送（その３）

35　この場合にはすべてのプロセス上で do 変数 jが１～ 100 の範囲，iが１～ 100 の範囲を処理
します。

ずれかです。listは基本整数型の１次元重複ローカル配列です。［と］で囲まれた部分は

省略できます。do変数を含む整数式では，例えば do変数を iとして，2*i+1 は正しいで

すが，1+2*iは誤りです。

３．添字の並び（i1,i2, . . . ,im, . . .）に同じ do変数を２回以上指定することはできません。

（ j1,j2, . . . , jn,. . .）も同様です。

４．添字の並び（i1,i2, . . . ,im, . . .）にはすべての do変数があらわれなければなりません。

（ j1,j2, . . . , jn,. . .）も同様です。

５　グローバル変数の resident 指定

resident指定はグローバル変数をローカル変数と同様に（高速に）アクセスできるようにする

機能です。resident指定されたグローバル変数はその変数をアクセスするプロセス上に割り当て

られている必要があります。すでにでてきた図７を resident指定で書き換えた図 14を使って以

下で説明します。

４～５行目 図７では分割ローカル配列 aを宣言しましたが，ここでは使用しません。

６行目 agを袖付きのグローバル配列として宣言します。

７行目 グローバル配列 agの袖部分の転送を行います。図 15に示すように agの袖（図

1 10 10 20  21 30 11  21

�� ag

 11 20

�� ag �� ag

����1 ����2 ����3

図 15　グローバル配列 ag の袖の転送

01: !xocl processor p(3)
02: !xocl index partition ind1=(p,index=1:30,part=band,overlap=(1,1))
03: !xocl index partition ind2=ind1(overlap=(0,0))
04:       dimension ag(30),b(30)
05: !xocl local b(/ind2)
06: !xocl global ag(/ind1)

　 　 　 　 　 　 　 　 …

07: !xocl overlapfix(ag) (id)
08: !xocl move wait (id)
09: !xocl spread do resident(ag) /ind2
10:       do i=2,29
11:         b(i)=ag(i-1)+ag(i)+ag(i+1)
12:       end do
13: !xocl end spread do

図 14　グローバル変数の resident 指定

15の網掛け部分）を対応する袖でない部分（図 15の斜線部分）で置き換えます。

例えば，プロセス２上の ag(10)の値はプロセス１上の ag(10)で置き換られ

ます。

９～ 13行目 ９行目の spread do文に resident指定をしています。これにより 10～ 12行目

の doループにあらわれる配列 agはローカル変数と同様の方法でアクセスする

ことができます。

resident指定すると分割ローカル配列と equivalence文で結合する必要がなくなるので，並列

プログラムへの書き換えの作業量が軽減される場合があります。resident指定は parallel region

文，spread region文，spread do文，spread move文，spread assignment文に指定できます。

また，resident文を使って宣言するとプログラム単位全体で有効となります。

６　重複ローカル配列の転送

同一の重複ローカル配列に対して，各プロセス上で別々の部分を定義したものを各プロセスか

らすべてのプロセスに転送するには unify文を使います。簡単な使用例を以下に示します。

!xocl processor p(2)

!xocl index partition ind= (p,index=1:10,part=band)

      dimension a(10,10)

　 　 　 　 　 　 　 　 …

!xocl spread do /ind

      do j=1,10

        do i=1,10

          a(i,j)=i*j

        end do

      end do

!xocl end spread do

!xocl unify(a(:,/ind))(id)

!xocl move wait(id)

この例では，重複ローカル配列 aに関して，doループにおいて a(1:10,1:5)の部分をプロセ

ス１で，a(1:10,6:10)の部分をプロセス２でそれぞれ定義しています。これは図 16中の転送

前の状態になります。つぎに，各プロセスで定義した部分を他のプロセスに転送するように分割

指定をして，unify文を実行します。unify文も転送の開始を指示する命令ですから，move wait

文によってデータ転送の完了を確認する必要があります。これによって，配列 aの全要素は各

プロセス上ですべて同じ値を持つことになります（図 16中の転送後の状態）。

７　プログラムの分割

プログラム中のある部分の処理を１つのプロセス（または複数のプロセス）上で行ない，プロ

グラムの別の部分の処理を別のプロセス上で行なう場合には spread region文を使います。つま

り，spread region文はプログラムをいくつかのリージョンに分割し，それぞれのリージョンを

１つまたは複数のプロセス 36に割り当てる機能を持つ文です。使い方は以下のようです。

!xocl processor p(10)

　 　 　 　 　 　 　 　 …

!xocl spread region /p(1)

　 　 　 　 　 　 　 　 … 　（処理 A）
!xocl end spread region

リージョンの分割は spread region文で始まり，end spread region文 37で終ります。この例で

は処理 Aをプロセッサグループ pの１番目のプロセス上で行なうことを指定しています。リー

ジョンを割り当てられなかったプロセス（この場合にはプロセス２～プロセス 10）はプロセス

１が処理Aを終えるまで待ちます 38。つぎの例では処理A，B，Cを並列処理するために３つのリー

ジョンに分割しています。

36　分割したリージョンの１つをさらに複数のリージョンに分割する場合などにはそのリージョン
に複数のプロセスを割り当てる必要があります。

37　end spread regionの regionは省略できます。
38　バリア同期を取ります。

図 16　unify 文による転送

!xocl processor p(10)

!xocl proc alias p456=p(4:6)
　 　 　 　 　 　 　 　 …

!xocl spread region /p(1)

　 　 　 　 　 　 　 　 … 　（処理 A）
!xocl region /p(2:3)

　 　 　 　 　 　 　 　 … 　（処理 B）
!xocl region /p456

　 　 　 　 　 　 　 　 … 　（処理 C）
!xocl end spread region

２つ目以降のリージョンの指定には region文を使います。プロセス１上では処理 Aを，プロセ

ス２及びプロセス３上では処理 Bを，プロセス４～プロセス６上では処理 Cを実行します。た

だし，proc alias文を使ってプロセッサグループ pの４～６台目の３つのプロセスにプロセッサ

グループ p456という別名を付けています。

８　配列式の並列化

単純 where文，配列式を使った代入文を並列処理するには spread assignment文を使います。

簡単な使用例を以下に示します。

!xocl processor p(10)

!xocl index partition ind=(p,index=1:100,part=band)

      dimension a(100),b(100),c(100)

!xocl local a(/ind),b(/ind),c(/ind)

　 　 　 　 　 　 　 　 …

!xocl spread assignment /ind

      where(a.ge.0.0)b=...

      where(a.lt.0.0)b=...

      c=a*b

!xocl end spread assignment

この例のように spread assignment文と end spread assignment文 39の間に並列処理したい単

純 where文や配列式を使った代入文を記述します。

39　end spread assignmentの assignmentは省略できます。

９　グローバル関数

doループを使って総和や最大値・最小値等を求めることはしばしば起こります。その doルー

プが分割されている場合に利用するのがグローバル関数です。グローバル関数には sum（総和），
max（最大値），min（最小値），and（論理積），or（論理和），last（最終値）40の６つがあり

ます。使い方はつぎのようになります。

!xocl spread do /ind

      do i=1,n

　 　 　 　 　 　 　 　 …

        arg =...

　 　 　 　 　 　 　 　 …

      end do

!xocl end spread do g-func(arg)[,g-func(arg)...]

g-funcの位置に６つのグローバル関数のいずれかを指定します。また，コンマで区切って複数の

グローバル関数を指定することもできます。引数 argには変数名または配列名が指定できますが，

いずれも重複ローカル変数でなければなりません。各プロセスに割り当てられた doループの繰り

返し範囲の処理が終了した時点でグローバル関数の引数の値がプロセス間で評価され，引数の値は

その結果に置き換えられます。引数が配列の場合には要素ごとに評価されて要素の値が結果に置き

換えられます。引数として指定できる型は，sumが整数型，実数型及び複素数型，maxと minが

整数型及び実数型，andと orが論理型，lastが整数型，実数型，複素数型及び論理型です。例

えば，分割ローカル配列 z（寸法 1000の１次元配列）の要素の総和を求めるにはつぎのようにし

ます。

      s=0.0d0

!xocl spread do /ind

      do i=1,1000

        s=s+z(i)

      end do

!xocl end spread do sum(s)

この doループを実行すると重複ローカル変数 sの値は配列 zの要素の総和になります。

グローバル関数 max及び minに引数として変数名を指定する場合には，最大８個までの引数

40　lastの引数として指定した重複ローカル変数について，最後のリージョンにおける値を返し
ます。例えば，ある定数と等しい値を持つ配列要素の添字の最大のものを探すときに使います。

を指定することができます。ただし，第２引数以下は整数型でなければなりません。第２引数以

下に関しては，プロセスの中で第１引数が最大値（または最小値）を与えたプロセス上の第２引

数以下の値がそのまま他のプロセス上の対応する引数に複写されます。例えば，分割ローカル配

列 zの要素の中の最大値と最小値及びその要素番号を求めるにはつぎのようにします。
      zmax=-1.0d5

      zmin= 1.0d5

!xocl spread do /ind

      do i=1,1000

        if(z(i).gt.zmax)then

          zmax=z(i)

          nmax=i

        endif

        if(z(i).lt.zmin)then

          zmin=z(i)

          nmin=i

        endif

      end do

!xocl end spread do max(zmax,nmax),min(zmin,nmin)

この場合には，各プロセスに割り当てられた doループの繰り返し範囲の処理が終了した時点で

グローバル関数 maxの第１引数（ここでは変数 zmax）の値がプロセス間で比較され，その最大

値を与えたプロセス上のすべての引数（ここでは変数 zmax及び nmax）の値が他のプロセス上

の対応する引数に複写されます。minに関しても同様です。

つぎの例はグローバル関数 sumを使って重複ローカル配列を転送するプログラムです。

!xocl processor p(2)

!xocl index partition ind=(p,index=1:10,part=band)

      dimension a(10,10)

　 　 　 　 　 　 　 　 …

      do j=1,10

        do i=1,10

          a(i,j)=0.0

        end do

      end do

!xocl spread do /ind

      do j=1,10

        do i=1,10

          a(i,j)=i*j

        end do

      end do

!xocl end spread do sum(a)

冗長実行で重複ローカル配列 aを０に初期化した後，spread do 文を使って a(1:10,1:5)の部

分をプロセス 1 で，a(1:10,6:10)の部分をプロセス２でそれぞれ定義します（図 17の総和を

とる前の状態）。つぎに sumの引数に配列 aを指定して，aの要素ごとにプロセス間で総和をと

ります。結果として，配列 aの全要素は各プロセス上で同じ値を持つことになります（図 17の

総和をとった後の状態）。

10　拡張 spread do 構文

拡張 spread do構文を使うと doループの繰り返しの１回ずつと任意のブロックをリージョン

として並列処理することができます。使用例を図 18に示します。これはすでにでてきた並列処

理プログラム例２のサブルーチン boundを，拡張 spread do構文を使って記述したものです。

以下で，図 18について説明します。

図 17　グローバル関数 sumによる重複ローカル配列の転送

１～ 12行目 この部分は図 10のサブルーチン diffusの宣言部分と同じです。

13行目 拡張 spread do構文の始まり。

14～ 17行目 この部分は do変数 jの値が０のときの処理を行います。このリージョンは１番

目のプロセスが処理します。

18～ 22行目 この部分は do変数 jの値に対応して割り当てられたそれぞれのプロセスが並列

処理します。

23～ 26行目 この部分は do変数 jの値が jmaxのときの処理を５番目（最後）のプロセスが

行います。

01:       subroutine bound(xmax,ymax,alpha,t)

02:       parameter (npe=5,imax= 100,jmax= 100)

03: !xocl processor mproc(npe)

04: !xocl subprocessor sproc=mproc(1:npe)

05: !xocl index partition ind1=(sproc,index=0:jmax, &

06: !xocl         part=band,overlap=(1,1))

07:       dimension u(0:imax,0:jmax),ug(0:imax,0:jmax)

08: !xocl local u(:,/ind1)

09: !xocl global ug

10:       equivalence (u,ug)

11:       common /data/ ug

12: !xocl index partition ind2=(sproc,index=0:jmax,part=band)

　　　　　　　　　 …

13: !xocl spread do /ind2

14: !xocl index 0

15:       do i=0,imax

16:         u(i,0)=...
17:       end do

18: !xocl index

19:       do j=1,jmax-1

20:         u(0,j)=...
21:         u(imax,j)=...
22:       end do

23: !xocl index jmax

24:       do i=0,imax

25:         u(i,jmax)=...
26:       end do

27: !xocl end spread do

28:       end

図 18　拡張 spread do 構文の例（サブルーチン bound）

11　ブロードキャスト

あるプロセス上の重複ローカル変数の値を他のすべてのプロセスに転送するには broadcast文

を使います。例えば，３番目のプロセス上で重複ローカル配列 aの各要素の値が定義されていて，

この値を他のプロセス上の配列 aに転送する場合 41には以下のようにします。

!xocl processor p(5)

      dimension a(100)

　 　 　 　 　 　 　 　 …

!xocl spread region /p(3)

      do i=1,100

        a(i)=...

      end do

!xocl end spread region

!xocl broadcast(a)(idvproc().eq.3)

サービス関数 idvprocは無引数の関数で，関数値として idvprocを呼び出したプロセスのプ

ロセッサ識別番号を返します 42。

12　排他制御

複数のリージョンで同じグローバル変数の値を更新する場合などには排他制御が必要になりま

すが，これには lockon文を使います。つぎの例では２つのリージョンでグローバル変数 k の値

をそれぞれ更新しています。

!xocl processor p(2)

!xocl global k

      k=0

!xocl parallel region

!xocl spread region /p(1)

!xocl   lockon 1

          k=k+1

!xocl   end lockon

!xocl region /p(2)

41　３番目のプロセス上の配列 aの値を他のプロセス上の配列 aにコピーします。
42　プロセッサ識別番号は 1 からプログラム開始時に割り当てられたプロセス数までのいずれかの
値を取ります。親プロセスのプロセッサ識別番号は１です。

!xocl   lockon 1

          k=k+1

!xocl   end lockon

!xocl end spread region

!xocl end parallel region

      write(6,*)k

同じ排他制御番号 43を持つ lockon文と end lockon文の間にある部分の実行はリージョン間で

排他制御されるので最後の write文で出力される kの値は（正しく）２になります。このプロ

グラムから lockon文と end lockon文を取り去ると write文で出力される値は不定になります。

lockon文はリージョン間の実行順序の制御はできません。

13　入出力

XPFortranの入出力文で扱うファイルには共有ファイルと専有ファイルの２種類があります。

共有ファイルはパラレルリージョン外でオープンされたファイルで，リージョンの内外を問わず

プログラムのどこからでもアクセスできます。専有ファイルはパラレルリージョンの中でオープ

ンされたファイルで，オープンしたリージョンの中だけ 44で入出力が可能です。ただし，ファ

イルのオープンは open文，read文，write文，backspace文，endfile文，print文のいずれか

が最初に実行されたときに行なわれます。標準入力ファイル（装置参照番号５），標準出力ファ

イル（装置参照番号６），標準エラー出力ファイルは共有ファイルです。

共有ファイルを異なるリージョンからアクセスする場合，入出力文の実行順序は不定になるの

で注意してください。

!xocl processor p(2)

      open(10,file=’a.data’)

　 　 　 　 　 　 　 　 …

!xocl spread region /p(1)

      write(10,*)i,j

!xocl region /p(2)

      write(10,*)x,y

!xocl end spread region

43　排他制御番号は１から 100 までの整数です。この例では 1 です。
44　ただし，そのリージョンの中でさらに子リージョンが生成される場合には子リージョンの部分
は除きます。

この例ではファイル a.dataを共有ファイルとして装置参照番号 10でオープンし，このファイ

ルにプロセス１上では変数 i及び jの値を，プロセス２上では変数 x及び yの値をそれぞれ出

力しています。入出力の結果は入出力文単位で保証されますからこの２組の数値が１つのレコー

ドに混在して出力されることはありませんが，２つの write文のどちらが先に実行されるかは不

定です。

異なるリージョンにある入出力文の実行順序を制御したい場合にはサービスサブルーチン
postevt及び waitevtを使ってつぎのようにします。

!xocl processor p(2)

      open(10,file=’a.data’)

　 　 　 　 　 　 　 　 …

!xocl spread region /p(1)

      write(10,*)i,j

      call postevt(1)

!xocl region /p(2)

      call waitevt(1)

      write(10,*)x,y

!xocl end spread region

postevtと waitevtは互いに同じ事象識別番号 45を引数にします。postevtが呼び出さ

れると事象識別番号に対応する事象が発生したことが他のプロセスに通知されます。一方，
waitevtを呼び出したリージョンに属するすべてのプロセス（この場合はプロセス２）では事

象識別番号に対応する事象が発生したという通知を受けとるまで待ち合わせします。したがって，

この例ではプロセス１上の write文が先に実行されます。postevtと waitevtの挿入位置に注

意してください。

冗長実行では通常はすべてのプロセス上で同一の処理が行なわれますが，入出力文に関しては

親プロセスのみが入出力処理を行ない，必要ならばプロセス間で通信を行なってその入出力処理

の結果が子プロセス上でも同一になるようにします。例えば，冗長実行時に read文を実行する

と親プロセスが入力処理を行ない，入力した値は子プロセスへ転送されてすべてのプロセス上で

同じ値になります。また，プロセスによって異なる値を持つ重複ローカル変数を出力する場合で

も，親プロセス上の変数の値のみが出力されるので注意してください。

入出力文の入出力並びに関する制約として，１つの入力文の入力並びの中にローカル変数とグ

ローバル変数を混在させることはできません。また，入力並び及び出力並びのいずれにも分割ロー

カル配列の配列名を指定することはできません 46。

45　事象識別番号は１から 100 までの整数です。この例では１です。
46　分割ローカル配列の配列要素名を指定することはできます。

14　プログラムミスの事例

これまでセンターの利用者の間で実際に起った XPFortranのプログラムミスの事例を以下で

紹介します。

14. １　事例１

つぎのような差分式の計算の並列化を考えます。

dimension a(30),b(30)

　 　 　 …

do i=2,29

  b(i)=a(i-1)+a(i)+a(i+1)

end do

配列を均等分割して 3 台のプロセスを使って計算することにすると，図 19のようなプログラム

になります。配列 aを袖付き分割ローカル配列，配列 bを袖なしの分割ローカル配列としてそ

れぞれ宣言し，aはグローバル配列 agと equivalence文で結合します。overlapfix文を使って

配列 aの袖部分を転送した後，spread do文を使って差分計算を分割して行ないます。

ところが，このプログラムを走らせてみると翻訳は正常終了するのですが，実行中に異常終了

したり，実行は正常終了しても正しい計算結果が得られなかったりします。実は，このプログ

!xocl processor p(3)

!xocl index partition ind1=(p,index=1:30,part=band,overlap=(1,1))

!xocl index partition ind2=ind1(overlap=(0,0))

      dimension a(30),ag(30),b(30)

!xocl local a(/ind1),b(/ind2)

!xocl global ag

      equivalence (a,ag)

　 　 　 　 　 　 　 　 …

!xocl overlapfix(a) (id)

!xocl movewait (id)

!xocl spread do /ind1

      do i=2,29

        b(i)=a(i-1)+a(i)+a(i+1)

      end do

!xocl end spread do

図 19　差分計算を誤って並列化したプログラム

ラムには spread do文の分割指定に誤りがあります。spread do文の分割指定に ind1（袖付き）

を指定すると，doループは１番目のプロセスでは do変数 iの値が２～ 11の範囲を，２番目の

プロセスでは 10～ 21の範囲を，３番目のプロセスでは 20～ 29の範囲をそれぞれ計算します

から，配列 a，b共に宣言した範囲以外の部分をアクセスすることになります。spread do文に

は袖なしの分割指定 ind2を指定してください。

14. ２　事例２

下のような分割された配列中の特定の要素にアクセスするプログラムを考えます。

      parameter(npe=16,imax=35)

!xocl processor p(npe)

!xocl index partition ind=(p,index=1:imax,part=band)

      dimension a(imax)

!xocl local a(/ind)

　 　 　 　 　 　 　 …

!xocl spread region /p(1)

        a(1)=...

!xocl region /p(npe)

        a(imax)=...

!xocl end spread region

プロセスを 16台使用して寸法 35の１次元ローカル配列 aを均等分割し，spread region文を使っ

て１番目のプロセス上で a(1)に，16番目のプロセス上で a(35)にそれぞれ値を代入していま

す。このように配列の端の部分だけにアクセスすることは境界条件を設定する場合などに必要と

なります。このプログラムも一見誤りはないようにみえるのですが，実際に走らせてみると正し

い結果は得られません。理由は a(35)に正しい値が代入されないためです。

配列 aの寸法 35をプロセス数 16で割ると商が２，余りが３となります。割り切れない時の

均等分割では，商に１を足した要素数を順にプロセスに割り当てていって，最後に残りの要素数

をそのつぎのプロセスに割り当てます。この場合には，１番目のプロセスに a(1)～ a(3)，２

番目のプロセスに a(4)～ a(6)，. . . ，11番目のプロセスに a(31)～ a(33)，12番目のプロ

セスに a(34)及び a(35)，と割り当てられます。したがって，13番目から 16番目までのプロ

セスには配列は割り当てられず，実質上「空回り」するだけになります。分割する配列の次元の

寸法とプロセス数との関係によっては必ずしも最後のプロセスに配列の最後の要素が割り当てら

れるわけではない，というわけです。同様な現象は上のプログラムで npeの値を 13，14，15と

したときにも起こります。npeの値が 12以下ならば正しく計算されますから，npeの値によっ

てうまくいったりいかなかったりしてわけがわからなくなります。

一般的には使用するプロセス数に比べて分割する配列の次元の寸法は十分大きい場合が多いの

で，このようなことが頻繁に起こるとは思われませんが，図 20に示したように，spread region 

文の代わりに spread do文と if文を使えばプロセス数や配列の寸法を意識せずに記述できます。

14. ３　事例３

下のように冗長実行中にグローバル変数にアクセスするプログラムを考えます。

!xocl processor p(3)

      dimension nn(3)

!xocl global nn

!xocl parallel region

      nn(idvproc())=idvproc()

!xocl end parallel region

      write(6,*)(nn(i),i=1,3)

 

プロセスを３台使用し，寸法３の１次元グローバル配列 nnを用意します。パラレルリージョ

ンに入ったところ，すなわち，冗長実行においてサービス関数 idvproc（呼び出したプロセス

のプロセッサ識別番号を関数値として返します）を使って配列 nnに値を代入し，その値を出

力します。関数値にしたがって，１番目のプロセスでは nn(1)に１を，２番目のプロセスでは

      parameter (npe=16,imax=35)

!xocl processor p(npe)

!xocl index partition ind=(p,index=1:imax,part=band)

      dimension a(imax)

!xocl local a(/ind)

　 　 　 　 　 　 　 …

!xocl spread do /ind

      do i=1,imax

        if(i.eq.1) then

          a(1)=...
        elseif(i.eq.imax) then

          a(imax)=...
        endif

      end do

!xocl end spread do

図 20　分割された配列中の特定の要素にアクセスするプログラム

nn(2)に２を，３番目のプロセスでは nn(3)に３をそれぞれ代入するように思われますが，残

念ながら write文の出力は，

1 0 0

となって期待どおりには動きません。このプログラムの代入文を if文を使ってつぎのように，

      if(idvproc().eq.1)then

        nn(1)=1

      elseif(idvproc().eq.2)then

        nn(2)=2

      else

        nn(3)=3

      endif

と書き換えてもやはりうまくいきません。これは，冗長実行では通常はすべてのプロセス上で同

一の処理が行なわれますが，グローバル変数へのアクセスに関しては動作が異なるからです。す

なわち，冗長実行ではグローバル変数の値を定義する（代入文の左辺にある）場合には親プロセ

スのみがその処理を行ないます 47。よって，各プロセスでグローバル変数に値を代入する場合に

は spread do文または spread region文を使う必要があります。例えば，spread do文を使って

プログラムを修正すると以下のようになります。

!xocl processor p(3)

      dimension nn(3)

!xocl global nn

!xocl parallel region

!xocl spread do /(p)

      do i=1,3

        nn(idvproc())=idvproc()

      end do

!xocl end spread do

!xocl end parallel region

      write(6,*)(nn(i),i=1,3)

47　冗長実行でグローバル変数を引用する（代入文の右辺にある）場合には，すべてのプロセスが
グローバル変数を引用したのと同じ結果になります。これは冗長実行中の入出力文の処理とよく
似ています。

14. ４　事例４

下のような spread do文を使って分割ローカル配列に代入するプログラムを考えます。

!xocl processor p(4)

!xocl index partition ind=(p,index=1:100,part=band)

      dimension a(10000,100),b(10000,100)

!xocl local a(:,/ind),b(:,/ind)

　 　 　 　 　 　 　 …

      do i=1,10000

!xocl spread do /ind

      do j=1,100

        b(i,j)=a(i,j)

      end do

!xocl end spread do

      end do

このプログラムには spread do文の外側に do変数 iに関するループがありますから spread do

文を 10000回実行することになります。すでに説明したように spread do文と end spread do文

の位置でバリア同期をとりますから，合計で 20000回のバリア同期が発生します。この例のよ

うにバリア同期が多発するプログラムを実行するとそのプログラムの経過時間や CPU時間が非

常に長くなる場合があります。実行そのものは正常に終了しますからプログラムミスとはいえな

いかもしれませんが，あまりタチの良くないプログラムといえます。

このプログラムの場合には spread do文と end spread do文の位置でバリア同期をとる必要は

ありませんから，

      do i=1,10000

!xocl spread nobarrier do /ind

      do j=1,100

        b(i,j)=a(i,j)

      end do

!xocl end spread nobarrier do

      end do

のように spread do文と end spread do文に nobarrier指定をすると経過時間や CPU時間が長

くなる現象を回避できます。もちろん，このプログラムの場合には

!xocl spread do /ind

      do j=1,100

        do i=1,10000

          b(i,j)=a(i,j)

        end do

      end do

!xocl end spread do

のようにして do変数 iに関するループを内側に入れてしまえばバリア同期が多発することはあ

りません。

Ⅲ．その他

１.　commonid文，barrier文，pass by local文

この３つの XPFortran構文については本稿では使い方を紹介しませんでした。以下に簡単な

説明を加えておきますが，具体的な使い方に関しては文献［１］をご覧ください。

• commonid文

overlapfix文，move wait文などに指定されるプロセッサ間データ転送識別名は，通常１つ

のプログラム単位の中だけで有効ですが，commonid文は２つのプログラム単位にまたがっ

て１つのプロセッサ間データ転送識別名が有効であることを宣言します。

• barrier文

barrier文が置かれた位置でバリア同期をとることを指定します。

• pass by local文

pass by local文の直後に置かれた call文中のサブルーチンの実引数としてあらわれるグロー

バル変数をローカル変数と同様の方法で引数渡しすることを指定します。

２.　組込み関数

組込み関数の実引数としてローカル変数，グローバル変数のいずれも用いることができます。

ただし，実引数としてグローバル変数を用いる場合には，通常アクセスに時間がかかるので処

理速度は遅くなります 48。

３.　サービスサブルーチン等

計時用サービスサブルーチン clock，clockm，gettodなどは呼び出したプロセス上での値

を返します。プロセスの問い合わせに関するサービス関数には novproc49，idvproc50など

48　resident指定や pass by local文の指定がある場合にはローカル変数と変わりません。
49　関数値としてプログラムの開始時に割り当てられた CPU 数を返します。このプロセス数は

processor文で宣言したプロセス数ではなく，ジョブの依頼時に割り当てた CPU 数になります。
計算機資源を有効に利用するため，ジョブの依頼時に割り当てた CPU 数と processor文で宣言し
たプロセス数とが必ず一致するように指定してください。

50　関数値として idvprocを呼び出したプロセスのプロセッサ識別番号を返します。プロセッサ識
別番号は１から novprocが返す値までのいずれかの値を取ります。親プロセスのプロセッサ識別
番号は１です。

があります。

４.　エラーメッセージ等

コンパイル時や実行時にシステムから出力されるエラーメッセージ，診断メッセージなどには

以下のようなものがあります。なお，文献［１，２］はセンター図書室で貸し出しをおこなっ

ています。また，センターの全国共同利用システムの利用者はセンターホームページ（URL

は http://www2.itc.nagoya-u.ac.jp）からオンラインマニュアルをみることもできます。

•並列化に関するコンパイルメッセージ

メッセージ idが jooで始まるものです。メッセージの説明は文献［１］を御覧ください。

•その他のコンパイルメッセージ

メッセージ idが jwdで始まるものです。メッセージの説明は文献［２］を御覧ください。

•実行時のメッセージ

メッセージ idが jweで始まるものです。メッセージの説明は文献［２］を御覧ください。

５.　利用法等に関する文献

本稿では紹介しなかった HPC2500の利用法等に関しては，以下の文献を参照してください。

• HPC2500の利用法

「スーパーコンピュータ及びアプリケーションサーバ利用の手引」をセンターホームページ

から参照してください。

• XPFortranプログラムのデバッグ支援・チューニング支援ツール

文献［１，３］を御覧ください。文献［３］もセンター図書室で貸し出しをおこなっています。

また，センターの全国共同利用システムの利用者はセンターホームページからオンラインマ

ニュアルをみることもできます。

６.　役に立つコンパイルオプション

•分割ローカル配列に関して，定義された領域以外をアクセスしていないかのチェックには

コンパイルオプション -Wxのサブオプション -Cfまたは -Crを使います。これらのオプ

ションをコンパイル時に指定することにより，実行時に領域外アクセスをチェックできます。
-Cfと -Crの違いは -Crの方が荒いチェックとなるため実行時のオーバーヘッドが -Cfの

場合に比べて削減されることです。いずれにしても実行に長時間を要するプログラムでは実

行時間が短くなるようにしてから指定した方がよいでしょう。以下に使用例を示します。

hpc% xpfrt -Wx,-Cr -o sample sample.f

分割ローカル配列に加えて重複ローカル配列のチェックをする場合にはサブオプションを
-Cfdまたは-Crdとします。redident指定したグローバル配列のチェックにはサブオプショ

ンを -Cfgまたは -Crgとします。

•プログラム中の parameter文で定義された定数の値を，プログラムを直接書き換えること

なく，コンパイルオプションによって変更することができます。使い方は以下のようです。
        xpfrt -Wx,-Pname=value ...



名古屋大学情報連携基盤センターニュース　Vol.５,　No.２－2006. 5－168

nameには parameter文中の定数名，valueには変更したい値を指定します。例えば，プロ

グラム中の parameter 文で使用プロセス数を npe=2と指定してあるとして，この値を４に

変更して翻訳するには以下のようにします。

hpc% xpfrt -Wx,-Pnpe=4 -o sample sample.f

